The error function

Katharine Long
Texas Tech University

- ## Definition of the error function

The error function erf(x) is defined by the definite integral

$$\operatorname{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2}\, dt.$$

This integral cannot be evaluated in closed form except for $x = 0$ and $x \to \pm\infty$. The values at those points are erf(0) = 0 and erf($\pm\infty$) = ± 1.

- ### Computing the error function in Mathematica

The function `Erf[x]` returns the error function. You can assume that it will be computed to high accuracy.

- ### Visualizing the error function

In[12]:=
```
Plot[{2/Sqrt[Pi] Exp[-x^2], Erf[x]}, {x, -4, 4}, PlotStyle → {Blue, Black},
 FrameLabel → {"x"}, PlotLegends → Placed[{"2/√π e^(-x^2)", "erf(x)"}, {Right, Bottom}],
 PlotLabel → "The error function and its derivative"]
```

Out[12]=



- ## Maclaurin series

The Maclaurin series for erf(x) is easily derived by integrating the Maclaurin series for the Gaussian. It is convergent for all $x \in \mathbb{C}$.

In[13]:=
```
S[n_, x_] := Normal[Series[Erf[x], {x, 0, n}]]
```

In[24]:=
```
s2[x_] = S[2, x];
```

In[25]:=
```
s4[x_] = S[4, x];
```

In[26]:=
```
s8[x_] = S[8, x];
```

In[20]:=
```
s16[x_] = S[16, x]
```

Out[20]=
$$-\frac{x^{15}}{37800\sqrt{\pi}}+\frac{x^{13}}{4680\sqrt{\pi}}-\frac{x^{11}}{660\sqrt{\pi}}+\frac{x^{9}}{108\sqrt{\pi}}-\frac{x^{7}}{21\sqrt{\pi}}+\frac{x^{5}}{5\sqrt{\pi}}-\frac{2x^{3}}{3\sqrt{\pi}}+\frac{2x}{\sqrt{\pi}}$$

In[30]:=
```
Plot[{Erf[x], s2[x], s4[x], s8[x], s16[x]}, {x, 0, 2},
 PlotLegends → Placed[
   {"erf(x)", "degree 1", "degree 3", "degree 7", "degree 15"}, {Left, Top}],
 FrameLabel → {"x", "erf(x)"},
 PlotLabel → "erf(x) and partial sums of its Maclaurin series"]
```

Out[30]=


Asymptotic expansion

The asymptotic expansion for erf(x) does not converge as $n \to \infty$ for any x. However, for fixed n, it converges as $x \to \infty$. It is an effective approximation to erf(x) when x is large.

In[31]:=
```
A[M_, x_] :=
 1 - Exp[-x^2] / Sqrt[Pi] / x (1 - Sum[(-1)^n (2 n - 1)!! / 2^n / x^(2 n), {n, 1, M}])
```

In[34]:=
```
a0[x_] = A[0, x]
```

Out[34]=
$$1-\frac{e^{-x^2}}{\sqrt{\pi}\,x}$$

In[73]:=
```
a4[x_] = A[4, x];
```

In[70]:=
```
a10[x_] = A[10, x]
```

Out[70]=
$$1-\frac{e^{-x^2}\left(\frac{1}{2x^2}-\frac{3}{4x^4}+\frac{15}{8x^6}-\frac{105}{16x^8}+\frac{945}{32x^{10}}-\frac{10395}{64x^{12}}+\frac{135135}{128x^{14}}-\frac{2027025}{256x^{16}}+\frac{34459425}{512x^{18}}-\frac{654729075}{1024x^{20}}+1\right)}{\sqrt{\pi}\,x}$$

In[72]:=
```
Plot[{Erf[x], a0[x], a4[x]}, {x, 1.2, 2.5}, PlotRange → {0.94, 1.001},
 FrameLabel → {"x", "erf(x)"}, PlotStyle → {Black, Blue, Red},
 PlotLegends → Placed[{"erf(x)", "1 term", "5 terms"}, {Center, Right}],
 PlotLabel → "erf(x) and asymptotic expansions"]
```

Out[72]=